
Subject: Re: color_quan - how for exactly 256 colors?
Posted by [JD Smith](#) on Fri, 17 Oct 2003 18:57:31 GMT
[View Forum Message](#) <> [Reply to Message](#)

On Fri, 17 Oct 2003 01:01:33 -0700, Oliver Thilmann wrote:

```
>> David Fanning <david@dfanning.com> wrote in message
>> news:<MPG.19f86c414eaf8be5989719@news.frii.com>...
>
>> What does this mean!? By definition, there will be no "exact" results
>> when you sample 16.7 million colors down to 256. It just, uh...,
>> mathematically can't be done. :-)
>
> What I mean is: I know that my image contains not more than 256
> different RGB colors (out of 16.7 million) - I created the RGB image
> from an indexed image and now I want to transform it back. This can be
> done exactly and I wondered whether IDL provides a method to get that
> done. Cheers,
```

Yes, with HISTOGRAM:

```
rgb_image=r+256L*(g+256L*b)
h=histogram(rgb_image,OMIN=om)
wh=where(h gt 0,cnt) # Should be fewer than 256
h[wh]=bindgen(cnt)
index_image=h[rgb_image]
colors=om+wh ; these are your <=256 colors
r_vec=colors AND 255L
g_vec=ishft(colors,-8) AND 255L
b_vec=ishft(colors,-16) AND 255L
tvlct,r_vec,g_vec,b_vec
tv,index_image
```

Probably not the most efficient method in the universe, given the sparseness of the histogram, but it gets the job done.

JD
